
Rozdział 6.

Funkcje tekstowe

Funkcje tekstowe służą w zasadzie do działań na tekstach, ale Excel jest tu tolerancyjny
i często argumentami mogą być także liczby. Przekonamy się o tym, analizując podane
w tym rozdziale przykłady. Dzięki funkcjom tekstowym można łatwo między innymi:
wyszukiwać w tekście wybrane znaki, zamieniać je na inne, pozbywać się zbędnych spa-
cji lub zamieniać małe litery na wielkie bądź odwrotnie.

Nie wszystkie funkcje działające na tekstach lub tekstów dotyczące zostały włączone do
tej kategorii. Na przykład, opisana w poprzednim rozdziale, funkcja CZY.TEKST została
zaliczona do funkcji informacyjnych. Również w innych kategoriach znajdują się funk-
cje działające na tekstach, lecz o nich, zgodnie z porządkiem tej książki, będzie mowa
w innych rozdziałach.

Podstawowe funkcje tekstowe obsługują teksty, w których znaki są pojedynczymi baj-
tami. Funkcje te mają swoje odpowiedniki przeznaczone do obsługi tekstów pisanych
znakami dwubajtowymi (znakami pełnej szerokości), co ma miejsce w takich językach
jak tajski. Ich nazwy są rozszerzone o kropkę oddzielającą i literę B. Na przykład odpo-
wiednikiem funkcji DŁ jest funkcja DŁ.B. Pierwsza zwraca liczbę znaków jednobajtowych
w tekście, a druga liczbę bajtów w tekstach zapisanych znakami dwubajtowymi. Ponie-
waż przydatność funkcji obsługujących teksty zapisane znakami dwubajtowymi jest
w Polsce niewielka, nie będę ich opisywał. Opiszę jedynie kilka funkcji napisanych
specjalnie dla języków ze znakami dwubajtowymi i nie mających odpowiedników obsłu-
gujących zwykłe teksty. Przykładem takiej funkcji jest np. BAHTTEXT.

ASC
Funkcja wbudowana, dostępna w VBA.

W językach z dwubajtowym zestawem znaków (DBCS) zmienia znaki o pełnej szero-
kości (dwubajtowe) na znaki o połówkowej szerokości (jednobajtowe).

Składnia:

 tekst — to tekst lub odwołanie do komórki zawierającej tekst. Tekst jest
zmieniany tylko wtedy, gdy zawiera znaki dwubajtowe — patrz rysunek 6.1
W przykładzie podanym w systemie pomocy Excel, słowo „EXCEL”, nie
zawierające znaków dwubajtowych, nie zostało zmienione.

Rysunek 6.1.
Działanie funkcji ASC

W VBA istnieje funkcja Asc zwracająca kod ASCII pierwszego znaku tekstu. Funkcja
VBA jest używana przez bezpośrednie użycie w procedurze; wywołanie arkuszowej
funkcji ASC musi być zgodne z metodą wywoływania funkcji arkuszowych, opisaną
we wstępie do części II.

BAHTTEXT
Funkcja wbudowana, dostępna w VBA.

Zamienia liczbę na tekst w języku tajskim, dodaje przyrostek „Baht”.

Składnia:

 liczba — to liczba lub odwołanie do komórki zawierającej liczbę, co
pokazuje przykład z systemu pomocy Excela pokazany na rysunku 6.2.

Rysunek 6.2.
Działanie funkcji
BAHTTEXT

Ciekawe, że odpowiedniej funkcji nie ma dla żadnego innego języka, nawet dla angiel-
skiego. John Walkenbach, jeden z najlepszych znawców Excela, przypuszcza, że w gru-
pie programistów tworzących ten program jest wielu miłośników kuchni tajskiej, którzy
w ten sposób wyrazili swą wdzięczność.

DŁ — LEN
Funkcja wbudowana, niedostępna w VBA.

Oblicza liczbę znaków ciągu tekstowego.

Składnia:

 tekst — tekst lub odwołanie do komórki zawierającej tekst.

Przykład 6.1. Wyróżnianie zbędnych spacji

Plagą wielu list, zwłaszcza otrzymywanych ze źródeł zewnętrznych, są dodatkowe spa-
cje; utrudniają np. sortowanie i grupowanie wyników, tworzenie sum pośrednich itd.,
bowiem dla Excela Jan Kowalski z jedną spacją między imieniem i nazwiskiem to cał-
kiem inna osoba niż Jan Kowalski z dwiema spacjami. Najtrudniejsze do wyszukania są
komórki zawierające dodatkowe spacje na końcu tekstu, gdyż takich spacji nie widzimy
przy przeglądaniu arkusza (np. w komórce A5 na rysunku 6.3)1. Ale wyszukanie nawet
widocznych spacji nadmiarowych nie jest możliwe, jeżeli danych są tysiące.

Należy pamiętać, że programy baz danych nie tolerują takich niedokładności i przed
eksportem przygotowanej listy do pliku, który ma być czytany przez inne oprogramo-
wanie, należy dane bezwzględnie oczyścić ze wszystkich śmieci, w tym z dodatkowych
spacji. Łatwiej to zrobić, jeżeli komórki z dodatkowymi spacjami wyróżnimy za pomocą
formatowania warunkowego, tak jak na rysunku 6.3.

Rysunek 6.3.
Użycie formatowania
warunkowego pozwala
wyróżnić komórki,
w których tekst zawiera
dodatkowe spacje

 

                                                          
1 Excel takie końcowe spacje traktuje dość tolerancyjnie i pomija np. przy automatycznym tworzeniu sum

pośrednich, ale dla spacji w środku tekstu nie jest wyrozumiały.

Zauważ, że w formule warunku formatowania zostały użyte adresy względne. W komórce
A5 tekst Anna Giza został wyróżniony z powodu dodatkowej spacji na końcu, widocznej
po ustawieniu kursora po przejściu do edycji zawartości komórki.

Tekst jest wyróżniany, jeżeli jego długość po usunięciu zbędnych spacji (w wyniku dzia-
łania funkcji USUŃ.ZBĘDNE.ODSTĘPY) jest różna od pierwotnej.

FRAGMENT.TEKSTU — MID
Funkcja wbudowana, niedostępna w VBA.

Zwraca fragment ciągu tekstowego — określoną liczbę znaków, począwszy od poda-
nej pozycji.

Składnia:

 tekst — tekst lub odwołanie do komórki z tekstem zawierającym znaki, które
należy wyodrębnić. Excel zezwala tu na użycie liczby jako pierwszego argumentu
funkcji, ale zawsze uwzględnia jedynie jej rzeczywistą wartość, a nie postać
wynikającą z formatowania (patrz rysunek 6.4).

Rysunek 6.4.
Funkcja FRAGMENT.TEKSTU
zawsze używa rzeczywistej wartości liczby,
nie uwzględniając formatowania

 liczba_początkowa — określa pozycję pierwszego znaku, który należy wyodrębnić
(numeracja rozpoczyna się od 1).

 liczba_znaków — liczba znaków, które należy wyodrębnić.

Jeżeli: funkcja zwraca

liczba_początkowa jest > od długości tekstu wartość "" (tekst pusty)
liczba_początkowa jest < od długości tekstu, ale liczba_początkowa
plus liczba_znaków jest ≥ od długości tekstu

znaki do końca tekstu

liczba_początkowa jest < niż 1 wartość błędu #ARG!
liczba_znaków jest < 0 wartość błędu #ARG!

Przykład 6.2. Wybieranie cyfr z liczby

Jako pierwszego argumentu funkcji FRAGMENT.TEKSTU można użyć liczby lub odwołania
do komórki zawierającej liczbę, jednakże otrzymany wówczas wynik może czasem zdzi-
wić niedoświadczonego użytkownika Excela. Przykład został pokazany na rysunku 6.4.

W obu komórkach B1 i B2 jest zapisana ta sama liczba, którą w komórce B2 pozosta-
wiono w formacie ogólnym, zaś w komórce B1 nadano jej format daty. Widać, że funk-
cja FRAGMENT.TEKSTU tego nie uwzględnia, zwracając w obu przypadkach ciąg tekstowy
(zauważ wyrównanie do lewej) złożony z 4. i 5. cyfry pobranej z zapisu ogólnego.

Przykład 6.3. Konwersja nietypowego formatu daty

Często Excela używamy do obróbki danych otrzymywanych z zewnętrznych źródeł, np.
z systemów komputerowych typu mainframe. Daty w takich plikach mogą być zapisane
w różnych formatach niezrozumiałych dla Excela, co wymusza stosowanie konwersji.

Przykładem mogą być: data i czas zapisane w formacie rrrrdddggmm, gdzie rrrr ozna-
cza rok, ddd kolejny dzień roku od 001 do 366, a gg i mm odpowiednio godziny i minuty
w formacie 24-godzinnym.

Rysunek 6.5 pokazuje sposób poradzenia sobie z tym problemem.

Rysunek 6.5.
Użycie funkcji
FRAGMENT.TEKSTU
do konwersji daty
i czasu zapisanych
w niestandardowym
formacie

Jak widać na rysunku 6.5, całkowita formuła konwersji ma w komórce B2 postać:
=DATA(LEWY(A2;4);1;1)
+FRAGMENT.TEKSTU(A2;5;3)
+CZAS(FRAGMENT.TEKSTU(A2;8;2);PRAWY(A2;2);0)

gdzie:

 LEWY(A2;4) pobiera cztery pierwsze cyfry (1999), co DATA(LEWY(A2;4);1;1)
zamienia na 36161, czyli numer kolejny daty 1 stycznia 1999 roku;

 FRAGMENT.TEKSTU(A2;5;3) wycina z liczby zapisanej w A2 trzy kolejne cyfry,
począwszy od piątej, tworząc tekst "245";

 dzięki „domyślności” Excela dodanie liczby i tekstu 36161+"245" daje liczbę
36 406;

 FRAGMENT.TEKSTU(A2;8;2) daje tekst "01" złożony z 8. i 9. cyfry, a PRAWY(A2;2)
tekst "12" złożony z cyfr 10. i 11., zaś CZAS("01";"12";0)=CZAS(1;12;0) daje
kolejną liczbę czasu 0,05;

 36 406+0,05=36 406,05 to kolejna liczba daty i czasu, która przez nadanie formatu
dd-mm-rrrr gg:mm jest widoczna w postaci 03-09-1999 01:12, podając
w zrozumiały sposób datę i godzinę.

KOD — CODE
Funkcja wbudowana, niedostępna w VBA.

Zwraca kod numeryczny pierwszego znaku ciągu tekstowego.

Składnia:

 tekst — tekst lub odwołanie do komórki zawierającej tekst.

Przykład 6.4. Użycie funkcji KOD do sprawdzania poprawności
wpisywanych danych

Często dane wpisywane do tabeli muszą spełniać określone warunki, być liczbami z od-
powiedniego zakresu albo określonymi ciągami znaków. Załóżmy, że tworzymy tabelę:
w pierwszej kolumnie znajduje się nazwa towaru, a w drugiej jej trzyznakowy kod,
rozpoczynający się od cyfry 5, z wielką literą na drugim miejscu (bez liter polskich)
i dowolną cyfrą na trzecim (np. 5A3 jest kodem prawidłowym, zaś 5ac, 6A3 i 5A33 nie
spełniają założeń).

Należy zabezpieczyć się przed wpisywaniem błędnych kodów, co zrobimy za pomocą
wbudowanej w Excela procedury sprawdzania poprawności. Funkcji KOD użyjemy do
sprawdzenia, czy na drugim miejscu w ciągu znakowym znajduje się wielka litera. Do-
zwolone litery to A, B…Z. Ich kody to odpowiednio: 65, 66…90 — łatwo to sprawdzić,
wpisując np. do B1 formułę =KOD(A1), a następnie do A1 litery, których kod chcemy
odczytać. Polskie litery Ą, Ć itd. mają kody spoza tego zakresu i ich uwzględnienie wyma-
gałoby rozbudowania, a więc wydłużenia formuły, a przy tym nie wniosłoby nic w istotę
rozumienia działania funkcji KOD.

 1. Zaznacz zakres, który ma być objęty sprawdzaniem — na rysunku 6.6 został
on wyróżniony innym kolorem — i wydaj polecenie Dane/Sprawdzanie
poprawności.

Rysunek 6.6.
Wpisywanie formuły
sprawdzania
poprawności kodu
wpisywanego
do kolumny B

 2. Na karcie Ustawienia wybierz Dozwolone/Niestandardowe, a w polu Formuła
wpisz formułę:

=ORAZ(
DŁ(B2)=3;
LEWY(B2;1)="5";
KOD(FRAGMENT.TEKSTU(B2;2;1))>64;KOD(FRAGMENT.TEKSTU(B2;2;1))<91;
CZY.LICZBA(WARTOŚĆ(FRAGMENT.TEKSTU(B2;3;1)))
)

 3. Na kartach Komunikat wejściowy i Alert o błędzie możesz dodatkowo wpisać
komunikaty objaśniające, np. takie jak na rysunku 6.7.

Rysunek 6.7.
Próba wpisania
błędnego kodu została
oprotestowana

Elementy formuły sprawdzające poszczególne znaki kodu zostały w punkcie 2. zapisane
w oddzielnych wierszach, aby ułatwić zrozumienie ich działania. Każdy pojedynczy ele-
ment formuły jest prosty, więc pomijam szczegółowe objaśnienia.

KWOTA — DOLLAR
Funkcja wbudowana, dostępna w VBA.

Zmienia liczbę na tekst i dodaje symbol waluty, przy czym następuje zaokrąglenie do
liczby miejsc dziesiętnych zgodnie z drugim argumentem funkcji. Stosowany format
to # ##0,00 zł.

Składnia:

 liczba — liczba lub odwołanie do komórki zawierającej liczbę (ew. formułę
zwracającą liczbę),

 miejsca_dziesiętne — liczba cyfr po przecinku dziesiętnym; pominięcie oznacza
wartość 2. Dopuszczalne jest podawanie wartości ujemnych, których znaczenie
pokazuje rysunek 6.8.

Rysunek 6.8.
Zaokrąglenia
sterowane drugim
argumentem funkcji
KWOTA są
wykonywane zgodnie
z zasadami arytmetyki

Należy zwrócić uwagę na różnicę między działaniem funkcji KWOTA a nadawaniem
liczbie formatu walutowego. Liczba, której nadano format walutowy, pozostaje liczbą
(na rysunku 6.8 w komórce A7 — wyrównanie do prawej), zaś funkcja KWOTA zamie-
nia liczbę na tekst (w komórkach A2:A6 — wyrównanie do lewej).

Przykład 6.5. Łączenie kwot walutowych z tekstem

Często musimy drukować rożnego rodzaju zestawienia, w których tekst łączymy z su-
mami pieniężnymi. Nadanie liczbie formatu walutowego zmienia jedynie sposób jej
wyświetlania, pozostawiając samą liczbę bez zmiany. Dlatego po połączeniu z tekstem
znak waluty nie jest widoczny, jak w komórce C2 w górnej części rysunku 6.9. Potrzebne
jest dołączenie dodatkowego członu tekstowego " zł", jak w komórce C3 na tym samym
rysunku (formuły użyte w komórkach C2 i C3 są pokazane obok w kolumnie E).

Rysunek 6.9.
Porównanie rysunków
górnego i dolnego
pokazuje, że funkcja
KWOTA ułatwia
łączenie tekstu
z sumami pieniężnymi

Dołączania dodatkowego członu, złożonego ze spacji i symbolu waluty, można uniknąć,
jeżeli użyjemy funkcji KWOTA, jak w dolnej części rysunku 6.9 w komórkach C2 i C3 (użyta
formuła jest widoczna w komórce E2).

LEWY — LEFT
Funkcja wbudowana, niedostępna w VBA.

Zwraca pierwszy znak lub pierwsze znaki ciągu tekstowego.

Składnia:

 tekst — tekst (ciąg tekstowy lub odwołanie do komórki zawierającej ciąg
tekstowy) zawierający znaki, które należy wyodrębnić.

 liczba_znaków — określa, ile znaków ma zwrócić funkcja LEWY. Liczba ta musi
być większa lub równa zeru. Jeśli liczba_znaków jest większa niż długość tekstu,
wynikiem funkcji LEWY jest cały tekst. Pominięcie argumentu liczba_znaków
jest równoznaczne z podaniem wartości 1.

Możliwości użycia funkcji LEWY zostały zaprezentowane w przykładach 6.3, 6.4, 6.6, 6.7
i 6.26. Jest często stosowana razem z funkcjami FRAGMENT.TEKSTU i ZNAJDŹ.

LITERY.MAŁE — LOWER
Funkcja wbudowana, niedostępna w VBA.

Zmienia w tekście litery wielkie na małe. Innych znaków nie zmienia.

Składnia:

 tekst — tekst (ciąg tekstowy lub odwołanie do komórki zawierającej ciąg
tekstowy).

Przykład 6.6. Tworzenie inicjałów pisanych małymi literami

Excel jest często używany jako narzędzie przygotowywania danych eksportowanych
później do baz danych, systemów księgowych itd. Niektóre takie systemy odróżniają
litery wielkie i małe i dla nich Kowalski i kowalski to dwie różne osoby, a e1JK i e1jk to
różne symbole. Dlatego w danych eksportowanych używanie liter wielkich bądź małych
powinno być ściśle kontrolowane.

Na podstawie listy pracowników i działów, w których pracują, należy utworzyć kody
pracowników. Każdy kod ma być pisany małymi literami i składać się z dwuznakowego
oznaczenia działu, pierwszej litery imienia i trzech pierwszych liter nazwiska. Rozwią-
zanie zostało pokazane na rysunku 6.10.

Rysunek 6.10.
Tworzenie kodów
pracowniczych
pisanych małymi
literami

Działanie formuły pokazanej na pasku edycji na rysunku 6.10 jest tak proste, że nie będę
go objaśniał.



152 Część II ♦ Funkcje

W kodach użyto trzech liter nazwiska, aby uniknąć powtarzania się kodów. Używanie
jedynie pierwszych liter imienia i nazwiska zwykle prowadzi do konfliktów, gdyż czę-
sto w jednej firmie, a nawet w jednym dziale pracują osoby, mające te same inicjały,
np. Jan Kowalski i Joanna Kącka.

LITERY.WIELKIE — UPPER
Funkcja wbudowana, niedostępna w VBA.

Zmienia w tekście litery małe na wielkie. Innych znaków nie zmienia.

Składnia:

 tekst — tekst (ciąg tekstowy lub odwołanie do komórki zawierającej ciąg
tekstowy).

Przykład 6.7. Sprawdzanie wprowadzonych danych. Modyfikacja
przykładu 6.4

W przykładzie 6.4 tworzyliśmy tabelę, której pierwsza kolumna zawierała nazwę towaru,
a drugiej — jej trzyznakowy kod, rozpoczynający się od cyfry 5, z wielką literą na dru-
gim miejscu (bez polskich liter) i dowolną cyfrą na trzecim (np. 5A3 jest kodem prawi-
dłowym, zaś 5ac, 6A3 i 5A33 nie spełniają założeń).

Teraz chcemy zezwolić, aby drugim znakiem kodu była po prostu litera, obojętnie, czy
wielka, czy mała. Wymaga to zmiany formuły sprawdzającej dane na następującą:

=ORAZ(
DŁ(B2)=3;LEWY(B2;1)="5";
KOD(LITERY.WIELKIE(FRAGMENT.TEKSTU(B2;2;1)))>64;KOD(LITERY.WIELKIE
(FRAGMENT.TEKSTU(B2;2;1)))<91;
CZY.LICZBA(WARTOŚĆ(FRAGMENT.TEKSTU(B2;3;1)))
)

Formuła FRAGMENT.TEKSTU(B2;2;1) „wykroi” z komórki drugi wpisany znak. Może to
być litera mała lub jakiś inny znak (w tym litera wielka). W pierwszym przypadku funk-
cja LITERY.WIELKIE zamieni małą literę na wielką, w drugim nie nastąpi żadna zmiana.
Kody liter wielkich należą do przedziału od 65 (kod A) do 90 (kod Z), więc sprawdze-
nie, czy kod znaku zwróconego przez funkcję LITERY.WIELKIE jest > 64 oraz < 91, jest
równoznaczne ze sprawdzeniem, czy znakiem wpisanym na drugim miejscu jest litera
(niezależnie od tego, czy jest mała, czy wielka).

Oczywiście, jak zwykle w Excelu, ta formuła mogła być napisana w inny sposób, ale
to pozostawiam domyślności czytelników.

OCZYŚĆ — CLEAN
Funkcja wbudowana, dostępna w VBA.

Usuwa z tekstu wszystkie znaki niedrukowane.

Składnia:

 tekst — tekst (ciąg tekstowy lub odwołanie do komórki zawierającej ciąg
tekstowy).

Przykład 6.8. Oczyszczanie importowanych danych

W danych importowanych ze źródeł zewnętrznych, np. z systemów mainframe lub z In-
ternetu, trafiają się znaki niedrukowalne (nierozpoznawalne), np. znaki o kodach z zakresu
1 – 31 oraz 129, 141, 143, 144 i 157. Często kilka takich różnych znaków Excel wyświe-
tla w arkuszu tak samo — np. znaki o kodach 11 i 129 są pokazywane jako kwadraciki,
widoczne na rysunku 6.11.

Rysunek 6.11.
Funkcja OCZYŚĆ usuwa
znaki o „złych” kodach,
tworząc zbitki, czasem
wymagające rozdzielenia,
jak na prawej części tego
rysunku

 

PODSTAW — SUBSTITUTE
Funkcja wbudowana, dostępna w VBA.

Zamienia w ciągu tekstowym stary_tekst na nowy_tekst.

Składnia:

 tekst — tekst (ciąg tekstowy lub odwołanie do komórki zawierającej ciąg
tekstowy),

 stary_tekst — to tekst lub odwołanie do komórki zawierającej tekst, którego
część znaków należy zastąpić,

 nowy_tekst — to tekst, który ma zastąpić stary_tekst,

 wystapienie_liczba — jeżeli stary_tekst występuje w tekst kilkakrotnie,
wystapienie_liczba określa, przy którym wystąpieniu dojdzie do zamiany.
Jeżeli ten argument jest pominięty, każdy znaleziony stary_tekst zostanie
zamieniony na nowy_tekst.

Funkcji PODSTAW należy używać, gdy określony ciąg znaków ma być zamieniony na inny,
np. formuła =PODSTAW("mama";"m";"t") zamienia m na t i daje tekst tata, przy czym miej-
sce wystąpienia m nie ma znaczenia.

Funkcja ZASTĄP służy do zamiany na inny dowolnego tekstu występującego w określo-
nym miejscu ciągu tekstowego, np. formuła =ZASTĄP("mama";3;2;"terac") dwa znaki
tekstu, zaczynając od trzeciego, zamienia na nowy tekst terac, przy czym istotne jest
miejsce rozpoczęcia zmiany i liczba zmienianych znaków, a nie ma znaczenia to, jakie
to są znaki.

Przykład 6.9. Zliczanie wystąpień wybranego znaku

Czasami chcemy sprawdzić, ilokrotnie wybrany znak występuje w tekście zapisanym
w komórce. Rysunek 6.12 przedstawia rozwiązanie.

Rysunek 6.12.
Zliczanie wystąpień
liter w tekście

Funkcja PODSTAW zamienia wybrany znak na tekst pusty, czyli praktycznie go usuwa
i zwraca tekst skrócony o tyle, ile wystąpień danego znaku zostało znalezionych. Odję-
cie długości tekstu skróconego od długości tekstu pełnego informuje, ile razy dany znak
został w tekście znaleziony.

Ponieważ funkcja podstaw odróżnia litery wielkie i małe, formuła w komórce A2 zliczyła
jedynie wystąpienia wielkiej litery Z, a formuła w A3 jedynie wystąpienia małej litery z.

W zmienionej formule w komórce A4
=DŁ($A$1)-DŁ(PODSTAW(LITERY.WIELKIE($A$1);"Z";""))

najpierw funkcja LITERY.WIELKIE zamienia wszystkie litery na wielkie, po czym wyszu-
kuje w tekście wielką literę Z. W tak zmienionym tekście wielkie litery Z są zamieniane
na pusty tekst. Różnica długości tekstu pierwotnego i skróconego odpowiada liczbie
znalezionych liter z (razem wielkich i małych).

Formuła w A5 działa analogicznie; również zlicza wszystkie litery z (wielkie i małe),
dokonując przed podstawieniem tekstu pustego zamiany wszystkich liter na małe.

PORÓWNAJ — EXACT
Funkcja wbudowana, niedostępna w VBA.

Porównuje dwa teksty i zwraca wartość PRAWDA, jeśli są takie same, lub FAŁSZ, gdy
są różne. Uwzględnia wielkość liter, ale ignoruje różnice w formatowaniu (patrz
rysunek 6.13).

Rysunek 6.13.
Porównanie
bezpośrednie komórek
a użycie funkcji
PORÓWNAJ

Składnia:

 tekst1, tekst2 — porównywane teksty lub odwołania do komórek
zawierających teksty (ciągi tekstowe).

Przykład 6.10. Zliczanie tekstów pisanych małymi (wielkimi literami)

W kolumnie A mamy zapisane kody jakichś operacji. Są to ciągi tekstowe złożone z liter
i cyfr. W kodach operacji przedpołudniowych użyto liter małych, a w kodach operacji
popołudniowych — liter wielkich. Naszym zadaniem jest policzenie operacji przedpo-
łudniowych i popołudniowych. Rozwiązanie zostało pokazane na rysunku 6.14.

Rysunek 6.14.
Użycie funkcji
PORÓWNAJ
do zliczania tekstów
pisanych małymi bądź
wielkimi literami

Formuła użyta w kolumnie B (pokazana na rysunku 6.14 na pasku edycji)
=JEŻELI(PORÓWNAJ(LITERY.MAŁE(A2);A2);1;0)

daje 1, gdy wszystkie litery występujące w tekście są małe.

Jeżeli w tekście w komórce A2:

 nie ma liter wielkich, oba teksty LITERY.MAŁE(A2) i A2 będą identyczne, więc
formuła PORÓWNAJ(LITERY.MAŁE(A2);A2) zwróci wartość PRAWDA — wówczas
funkcja JEŻELI zwróci wartość drugiego argumentu, czyli 1;

 są litery wielkie, to teksty LITERY.MAŁE(A2) i A2 będą różne i formuła
PORÓWNAJ(LITERY.MAŁE(A2);A2) zwróci wartość FAŁSZ, a funkcja JEŻELI
wartość trzeciego argumentu, czyli 0.

W analogiczny sposób można przeanalizować działanie formuły
=JEŻELI(PORÓWNAJ(LITERY.WIELKIE(A2);A2);1;0)

użytej w kolumnie C do zliczania tekstów zapisanych wielkimi literami.

POWT — REPT
Funkcja wbudowana, dostępna w VBA.

Tworzy ciąg tekstowy, powtarzając podany tekst określoną liczbę razy.

Składnia:

 tekst — tekst, który ma być powtarzany,

 ile_razy — liczba dodatnia określająca liczbę powtórzeń tekstu.

Jeśli argument ile_razy

 ma wartość 0, funkcja POWT zwraca "" (pusty tekst),

 nie jest liczbą całkowitą, jest do takiej liczby obcinany2.

Wynik funkcji POWT nie może być dłuższy niż 32 767 znaków. W przeciwnym wypadku
funkcja zwróci wartość błędu #ARG!.

Przykład 6.11. Tworzenie prostego histogramu

Funkcji POWT można użyć do prostej graficznej prezentacji danych, co zostało pokazane
na rysunku 6.15.

Rysunek 6.15.
Użycie funkcji POWT
do prostej graficznej
prezentacji danych

                                                          
2 Obcinany, a nie zaokrąglany zgodnie z regułami arytmetyki, np. ile_razy = 3,7 spowoduje trzykrotne

powtórzenie tekstu (przy zaokrągleniu byłyby to 4 powtórzenia).

PRAWY — RIGHT
Funkcja wbudowana, niedostępna w VBA.

Zwraca ostatni znak lub znaki wyodrębnione z ciągu tekstowego.

Składnia:

 tekst — ciąg tekstowy lub odwołanie do komórki zawierającej ciąg tekstowy,
zawierający znaki, które mają być wyodrębnione.

 liczba_znaków — liczba znaków, jakie ma zwrócić funkcja PRAWY. Argument
liczba_znaków musi być większy lub równy zeru. Argument opcjonalny
domyślnie równy 1.

Jeśli argument liczba_znaków jest większy od długości tekstu, to funkcja PRAWY zwraca
cały tekst.

O zastosowaniu funkcji PRAWY była już mowa w przykładzie 4.3.

SZUKAJ.TEKST — SEARCH
Funkcja wbudowana, dostępna w VBA.

Zwraca pozycję (numer znaku, licząc od lewej), na której po raz pierwszy szukany
znak lub ciąg znakowy (szukany_tekst) wystąpił w tekście przeszukiwanym (obej-
mujący_tekst).

Od funkcji ZNAJDŹ różni się nieodróżnianiem liczb wielkich i małych oraz dopuszczeniem
do stosowania znaków zastępczych.

Składnia:

 szukany_tekst — tekst, którego szukamy, lub odwołanie do komórki zawierającej
tekst szukany,

 obejmujący_tekst — tekst (odwołanie do komórki zawierającej tekst), w którym
szukany_tekst jest szukany,

 liczba_początkowa — numer znaku w argumencie obejmujący_tekst, od którego
ma się zacząć przeszukiwanie. Pominiecie oznacza użycie domyślnej wartości 1.

asedzielewski
Prostokąt


